
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 15.3.4 which is part of the SSNITZ test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_15_3_4
Test Group:
/15/CLIP_CLIR_CNAP.ttcn
ATS Version:
iwd-TTCN3-B2015-03_D15wk50 + essential modifications.

System Simulator used:
Anite Conformance Toolset Solution

UE used:
Intel 7360
Verification Status:
PASS
4 Corrections required for test case 15.3.4
Introduction

This section describes the changes required to make test case 15.3.4 run correctly with a 3G UE. All modifications are marked with label “@Anite@” in the TTCN comments.

The ATS version used as basis was SSNITZ test suite which is part of iwd-TTCN3-B2015-03_D15wk38 release. This test suite provided by MCC160 contains UTRAN test cases.

Function f_TC_15_3_4_UTRAN
	Record Name
	function f_TC_15_3_4_UTRAN 

	Reason for change
	1. At step 2 and 10, According to the 25.331, the uplink direct transfer procedure shall be initiated when the upper layers request a transfer of a NAS message on an existing signalling connection. When not stated otherwise elsewhere, the UE may initiate the uplink direct transfer procedure when another procedure is ongoing, and in that case the state of the latter procedure shall not be affected.
2. Fetching the start values wrongly during the security mode procedure.
3. In step 5 and 13, Release Complete message, Protocol discriminator should be non call related SS and the constraint cs_FacilityIE_lv omits the IE identifier ‘iei” field. As per 24.080 clause 3.5 facility should contain “1c”, which is omitted in cs_FacilityIE_lv constraint.
4. While Disconnect after test body, according to spec 24007 clause 11.2.3.1.3 the calue of flag in the Transaction Identifier sould be set to 1 as the transaction is initiated by the receiver.
5. In Function ‘f_UT_TriggerCNAPResponse’ is the command CNAP_DUMMY is not waiting for the confirmation although the confirmation is set as CNF_REQUIRED. While sending this command the TTCN set the CoordPortIndex to 1 in function fl_UT_HandleCmd() and when TTCN tries to send the next command it first checks if there is any AT or MMI command ongoing. At that point the test case gives error as “unexpected Upper Tester command (other command is still pending)” as the CNAP_DUMMY is still pending.

	Summary of change
	1. At step 2 and 10, used uplink direct transfer instead if initial direct transfer.
2. Removed the step to fetch Start values.
3. In step 5 and 13, as per above reason Protocol discriminatior ‘tsc_pd_ss’ is used and cs_FacilityIE_tlv is used instead of cs_FacilityIE_lv.
4. Changed the arguments in function f_UTRAN_CC_Disconnect as per the reason above.
5. Commented function ‘f_UT_TriggerCNAPResponse’.

	Source of change
	/15/CLIP_CLIR_CNAP.ttcn


Before
  function f_TC_15_3_4_UTRAN() runs on UTRAN_PTC

  { /* CNAP/Interrogation rejected */

    var octetstring v_EncodedComponents;

    var octetstring v_Octetstring;

    var InvokeIdType v_InvokeId := 0;

    var START_Value v_START_Value := '00000000000000000000'B;

    var Components v_Components;

    var TI v_TI_S, v_TI_R;

    var RRC_DATA_IND v_RRC_DataInd;

    //Initialise all UTRAN cells according to 34.108

    f_UTRAN34_Init();

    // Cell1 started switched on with default power level

    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);

    f_UTRAN34_SendDefSysInfo(utran34_Cell1);

    // Setup MO CS call  34.108 cl. 7.2.3.2

    f_UTRAN34_Preamble(utran34_Cell1);

    f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); // @sic R5-145740 sic@

    f_UTRAN_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // The UE is made to initiate an interrogation for CNAP (all).

    f_UT_SupplementaryService_CNAP(UT, QUERY_MODE);

    // @siclog "Step 2" siclog@

    // The UE sends CM SERVICE REQUEST cause: "supplementary service activation"

    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_CMServReq_SSDef(?)))-> value v_RRC_DataInd;
    v_START_Value := v_RRC_DataInd.ttcn_start;

    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);
    // Security  procedure

    f_UTRAN34_RRC_Security(utran34_Cell1, false, cs_domain);
    // @siclog "Step 3" siclog@

    // CM SERVICE ACCEPT

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                                 tsc_RB3,

                                 cs_U_CM_ServAcc));

    // @siclog "Step 4" siclog@

    // Receive REGISTER message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                            tsc_RB3,

                                            cr_U_RegisterMO(cr_FacilityIE_tlv,

                                                             cr_SS_VersionIndAny))) -> value v_RRC_DataInd;

    v_Octetstring := v_RRC_DataInd.msg.register.facility_Str.components;

    v_Components := f_SS_ComponentsDecode(v_Octetstring);

    if (not match(v_Components, cr_Components(cr_InvokeComponent(cr_InvokeInterrogate(?, cr_SS_ForBS_Code('19'O))))))  {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Components do not match to template");

    }

    // Get invokeID sent by the UE, to be used in SS response

    v_InvokeId := v_Components[0].invoke.invokeID;

    // Set TI value for response to be sent from the SS

    v_TI_R := v_RRC_DataInd.msg.register.ti;

    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);

    // Set up the upper tester to receive an unsolicited response

    f_UT_TriggerCNAPResponse (UT, int2str(tsc_ErrorCode_SSNotAvailable)); //  @sic R5-153746 sic@
    // @siclog "Step 5" siclog@

    // Send RELEASE COMPLETE message Return_error = SS not available

    v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode (v_InvokeId, tsc_ErrorCode_SSNotAvailable))));

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                               tsc_RB3,

                               cs_U_ReleaseComplete(v_TI_S, -, cs_SS_Cau(tsc_SS_CauValue_RequestedFacilityNotSubscribed), cs_FacilityIE_lv(v_EncodedComponents))));
    // @siclog "Step 6" siclog@

    // The UE shall display the response from SS

    if (f_UT_ChkResponseDisplayed(UT, int2str(tsc_ErrorCode_SSNotAvailable))) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    // @siclog "Step 7-8" siclog@

    // STATUS ENQUIRY - STATUS: CC state U10

    f_UTRAN34_CheckStatus(v_TI_S,  v_TI_R,  "Step 8", -, tsc_CallState_U10);

    // @siclog "Step 9" siclog@

    // The UE is made to initiate an interrogation for CNAP (all).

    f_UT_SupplementaryService_CNAP(UT, QUERY_MODE);

    // @siclog "Step 10" siclog@

    // The UE sends CM SERVICE REQUEST cause: "supplementary service activation"

    U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated,

                                          tsc_RB3,

                                          cr_U_CMServReq_SSDef(?)))-> value v_RRC_DataInd;
    v_START_Value := v_RRC_DataInd.ttcn_start;

    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);

    // @siclog "Step 11" siclog@

    // CM SERVICE ACCEPT

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                                 tsc_RB3,

                                 cs_U_CM_ServAcc));

    // @siclog "Step 12" siclog@

    // Receive REGISTER message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                            tsc_RB3,

                                            cr_U_RegisterMO(cr_FacilityIE_tlv,

                                                             cr_SS_VersionIndAny))) -> value v_RRC_DataInd;

    v_Octetstring := v_RRC_DataInd.msg.register.facility_Str.components;

    v_Components := f_SS_ComponentsDecode(v_Octetstring);

    if (not match(v_Components, cr_Components(cr_InvokeComponent(cr_InvokeInterrogate(?, cr_SS_ForBS_Code('19'O))))))  {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Components do not match to template");

    }

    // Get invokeID sent by the UE, to be used in SS response

    v_InvokeId := v_Components[0].invoke.invokeID;

    // Set TI value for response to be sent from the SS

    v_TI_R := v_RRC_DataInd.msg.register.ti;

    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);

    // Set up the upper tester to receive an unsolicited response

    f_UT_TriggerCNAPResponse (UT, int2str(3));  //  @sic R5-153746 sic@
    // @siclog "Step 13" siclog@

    // Send RELEASE COMPLETE message Invoke_problem = Resource Limitation

    v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnReject(cs_Reject_ResourceLimitation(v_InvokeId)))); //Problem resourceLimitation

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_ReleaseComplete(v_TI_S, -, cs_SS_Cau(tsc_SS_CauValue_ResourcesUnavailableUnspecified), cs_FacilityIE_lv(v_EncodedComponents))));
    // @siclog "Step 14" siclog@

    // The UE shall display the response from SS

    if (f_UT_ChkResponseDisplayed(UT, int2str(3))) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");

    }

    // @siclog "Step 15-16" siclog@

    // STATUS ENQUIRY - STATUS: CC state U10

    f_UTRAN34_CheckStatus(v_TI_S,  v_TI_R,  "Step 16", -, tsc_CallState_U10);

    f_UTRAN_TestBody_Set(false);

    f_UTRAN_CC_Disconnect(utran34_Cell1, cs_TI_MT); // @sic R5s140611 sic@
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);

    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));

  } // end of f_TC_15_3_4_UTRAN
After
function f_TC_15_3_4_UTRAN() runs on UTRAN_PTC

  { /* CNAP/Interrogation rejected */

    var octetstring v_EncodedComponents;

    var octetstring v_Octetstring;

    var InvokeIdType v_InvokeId := 0;

    var START_Value v_START_Value := '00000000000000000000'B;

    var Components v_Components;

    var TI v_TI_S, v_TI_R;

    var RRC_DATA_IND v_RRC_DataInd;

    //Initialise all UTRAN cells according to 34.108

    f_UTRAN34_Init();

    // Cell1 started switched on with default power level

    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);

    f_UTRAN34_SendDefSysInfo(utran34_Cell1);

    // Setup MO CS call  34.108 cl. 7.2.3.2

    f_UTRAN34_Preamble(utran34_Cell1);

    f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); // @sic R5-145740 sic@

    f_UTRAN_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // The UE is made to initiate an interrogation for CNAP (all).

    f_UT_SupplementaryService_CNAP(UT, QUERY_MODE);

    // @siclog "Step 2" siclog@

    // The UE sends CM SERVICE REQUEST cause: "supplementary service activation"

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated, //@Anite@

                                          tsc_RB3,

                                          cr_U_CMServReq_SSDef(?)))-> value v_RRC_DataInd;
    //v_START_Value := v_RRC_DataInd.ttcn_start; //@Anite@

    //f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value); //@Anite@
    // Security  procedure

    f_UTRAN34_RRC_Security(utran34_Cell1, false, cs_domain);

    // @siclog "Step 3" siclog@

    // CM SERVICE ACCEPT

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                                 tsc_RB3,

                                 cs_U_CM_ServAcc));

    // @siclog "Step 4" siclog@

    // Receive REGISTER message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                            tsc_RB3,

                                            cr_U_RegisterMO(cr_FacilityIE_tlv,

                                                             cr_SS_VersionIndAny))) -> value v_RRC_DataInd;

    v_Octetstring := v_RRC_DataInd.msg.register.facility_Str.components;

    v_Components := f_SS_ComponentsDecode(v_Octetstring);

    if (not match(v_Components, cr_Components(cr_InvokeComponent(cr_InvokeInterrogate(?, cr_SS_ForBS_Code('19'O))))))  {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Components do not match to template");

    }

    // Get invokeID sent by the UE, to be used in SS response

    v_InvokeId := v_Components[0].invoke.invokeID;

    // Set TI value for response to be sent from the SS

    v_TI_R := v_RRC_DataInd.msg.register.ti;

    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);

    // Set up the upper tester to receive an unsolicited response

    //f_UT_TriggerCNAPResponse (UT, int2str(tsc_ErrorCode_SSNotAvailable)); //  @sic R5-153746 sic@ @Anite@
    // @siclog "Step 5" siclog@

    // Send RELEASE COMPLETE message Return_error = SS not available

    v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode (v_InvokeId, tsc_ErrorCode_SSNotAvailable))));

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                               tsc_RB3,

                               cs_U_ReleaseComplete(v_TI_S, tsc_PD_SS, cs_SS_Cau(tsc_SS_CauValue_RequestedFacilityNotSubscribed), cs_FacilityIE_tlv(v_EncodedComponents)))); //@Anite@
    // @siclog "Step 6" siclog@

    // The UE shall display the response from SS

    if (f_UT_ChkResponseDisplayed(UT, int2str(tsc_ErrorCode_SSNotAvailable))) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    // @siclog "Step 7-8" siclog@

    // STATUS ENQUIRY - STATUS: CC state U10

    f_UTRAN34_CheckStatus(v_TI_S,  v_TI_R,  "Step 8", -, tsc_CallState_U10);

    // @siclog "Step 9" siclog@

    // The UE is made to initiate an interrogation for CNAP (all).

    f_UT_SupplementaryService_CNAP(UT, QUERY_MODE);

    // @siclog "Step 10" siclog@

    // The UE sends CM SERVICE REQUEST cause: "supplementary service activation"

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,  //@Anite@

                                          tsc_RB3,

                                          cr_U_CMServReq_SSDef(?)))-> value v_RRC_DataInd;
    v_START_Value := v_RRC_DataInd.ttcn_start;

    f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value);

    // @siclog "Step 11" siclog@

    // CM SERVICE ACCEPT

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                                 tsc_RB3,

                                 cs_U_CM_ServAcc));

    // @siclog "Step 12" siclog@

    // Receive REGISTER message

    U_Dc.receive(car_UplinkDirectTransfer(utran_CellDedicated,

                                            tsc_RB3,

                                            cr_U_RegisterMO(cr_FacilityIE_tlv,

                                                             cr_SS_VersionIndAny))) -> value v_RRC_DataInd;

    v_Octetstring := v_RRC_DataInd.msg.register.facility_Str.components;

    v_Components := f_SS_ComponentsDecode(v_Octetstring);

    if (not match(v_Components, cr_Components(cr_InvokeComponent(cr_InvokeInterrogate(?, cr_SS_ForBS_Code('19'O))))))  {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Components do not match to template");

    }

    // Get invokeID sent by the UE, to be used in SS response

    v_InvokeId := v_Components[0].invoke.invokeID;

    // Set TI value for response to be sent from the SS

    v_TI_R := v_RRC_DataInd.msg.register.ti;

    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);

    // Set up the upper tester to receive an unsolicited response

    //f_UT_TriggerCNAPResponse (UT, int2str(3));  //  @sic R5-153746 sic@ @Anite@
    // @siclog "Step 13" siclog@

    // Send RELEASE COMPLETE message Invoke_problem = Resource Limitation

    v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnReject(cs_Reject_ResourceLimitation(v_InvokeId)))); //Problem resourceLimitation

    U_Dc.send(cas_DataReq_CS(utran_CellDedicated,

                             tsc_RB3,

                             cs_U_ReleaseComplete(v_TI_S, tsc_PD_SS, cs_SS_Cau(tsc_SS_CauValue_ResourcesUnavailableUnspecified), cs_FacilityIE_tlv(v_EncodedComponents)))); //@Anite@
    // @siclog "Step 14" siclog@

    // The UE shall display the response from SS

    if (f_UT_ChkResponseDisplayed(UT, int2str(3))) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");

    }

    // @siclog "Step 15-16" siclog@

    // STATUS ENQUIRY - STATUS: CC state U10

    f_UTRAN34_CheckStatus(v_TI_S,  v_TI_R,  "Step 16", -, tsc_CallState_U10);

    f_UTRAN_TestBody_Set(false);

    f_UTRAN_CC_Disconnect(utran34_Cell1, v_TI_S); // @sic R5s140611 sic@ @Anite@
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);

    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));

  }

5 Branches executed in test case 15.3.4
The test case implementation was executed with Integrity and ciphering enabled.

The Intel 7360 passed this test case on Anite Conformance Testing Solution. The documentation below is enclosed as evidence of the successful test case run [1]:
6 Execution Log Files

Intel 7360 Handset

The Intel Handset model ‘Intel 7360’ passed this test case on Anite Conformance Toolset Solution. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: TC_15_3_4_UE1_LOG.html
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References

